

CLAZY: Lazy Calling for Common Lisp

Marco Antoniotti

Lab Linguaggi Programmazione – Programming Language Lab < LLP ΠΛΛ >

Dipartimento di Informatica, Sistemistica e Comunicazione,

Università Milano Bicocca

U14 - Viale Sarca 336, I-20126 Milan, ITALY

Abstract

This document contains a description of a **Common Lisp** extension that allows a programmer to write functional programs that use *normal order* evaluation, as in *non-strict* languages like Haskell. The extension is relatively straightforward, and it appears to be the first one such that is integrated in the overall **Common Lisp** framework¹.

1 Introduction

Common Lisp is a functional language (and also an imperative, object-oriented one, which, moreover, can be used in a declarative fashion). As a functional language it falls in the category of *strict* languages like ML [7] and OCaml [5], unlike Haskell [6], which is in the category of *normal-order* or *lazy* languages.

That is to say that the following code will enter an infinite loop, should it be executed at the **Common Lisp** prompt.

```
cl-prompt> (defun si (condicio ergo alternatio)
             (if condicio ergo alternatio))
```

```
SI
```

```
cl-prompt> (si t 42 (loop))
```

In a *lazy* language the function `si` (if in Latin) would return 42 instead of waiting for the form `(loop)` to produce a value.

In a bout of Haskell envy, I decided to look into some extensions to **Common Lisp** that would introduce ways to program in a lazy way. The result may sound *crazy*, and, in fact, a little bit it is.

The notion of *lazy evaluation* dates back to the Algol days and the notion of *by-name* parameter passing. In the Lisp camp, the best known way to introduce a form of lazy evaluation is to implement *streams* as described in *Structure and Interpretation of Computer Programs* (SICP) [1]; incidentally this form of lazy evaluation is also used by Okasaki [8] in his exposition of *functional data structures* in ML.

In SICP, streams are implemented using two primitives, `force` and `delay`, which can then be used to build a lazy container (the “stream”) using a *macro* `cons-stream`, and two accessors `head` and `tail`. A sufficient implementation in **Common Lisp** is the following:

¹A version of this note was presented at the 1st European Lisp Symposium 2008, Bordeaux, France

```
(defmacro delay (expr) `(lambda () ,expr))

(defun force (thunk) (funcall thunk))

(defmacro cons-stream (head tail) `(cons ,head (delay ,tail)))

(defun head (lazy-stream) (car lazy-stream))

(defun tail (lazy-stream) (force (cdr lazy-stream)))
```

At this point there are several Common Lisp packages floating around the net, that implement this flavor of lazy evaluation. E.g., *Heresy* [4], *funds* [2] and *FSet* [3] are exemplars of this approach. CLAZY goes off a (different) tangent and provides a more fundamental way to build such lazy constructions.

1.1 Limits of the delay/force Duo

Given `delay` and `force`, one could always implement the operator `si` as a macro using `delay`, as in

```
(defmacro si (condicio ergo alternatio)
  `(if (force ,condicio)
      (force (delay ,ergo))
      (force (delay ,alternatio))))
```

but this is a bit unsatisfactory as far as Haskell envy is concerned. `si` cannot be funcalled in any meaningful way and cannot be passed around as we would expect a regular function to be. A different solution is needed.

2 Defining and Calling Lazy Functions

It is possible to come up with a more satisfactory solution that will allow us to bypass `delay` and `force`, at the price of tweaking the “calling convention”. Then we can write `si` as:

```
(deflazy (condicio ergo alternatio)
  (if condicio ergo alternatio))
```

where `deflazy` defines both *lazy* and *strict* versions of the operator. The *lazy function* `si` can now be called as

```
CL prompt> (lazy:call #'si t 42 (loop))
42
```

```

CL prompt> (lazy:call (lazy #'(lambda (condicio ergo alternatio)
                             (if condicio
                                 ergo
                                 alternatio)))
                t
                (+ 20 20 2)
                (loop))

```

42

Figure 1: An example of the use of the special operator `lazy`.

I.e., `lazy:call` is the lazy version of `funcall`. The complexity of writing lazy code is thus moved to the call points. This may or may not be desirable, but it can be argued that this is a slightly better way than having to manually `force` expressions. In any case, the CLAZY approach still uses the `delay/force` duo under the hood, and they are available for more manual intervention.

From the example above, it should be apparent that `lazy:call` is a macro that does something special with the call, recognizing functions that are defined via `deflazy`. As a matter of facts, the expansion of `lazy:call` looks like this:

```

(lazy:call <op> <arg1> <arg2> ... <argN>)
=>
(funcall <lazyfied op>
        <thunked arg1>
        <thunked arg2>
        ...
        <thunked argN>)

```

The “lazy” version of `<op>` is defined by `deflazy` and each `<thunked argi>` is a closed over version of the argument as if `delay` was invoked on it.

Of course, a simple version of such idea can be easily implemented with a few macros, however, a well integrated version within the overall Common Lisp environment requires a few more bits and pieces. As example, CLAZY wants to make the analogy between `lazy:call` and `funcall` as tight as possible. This means that we need a way to pass (almost) regular `lambda`'s to `lazy:call`. This can be done the special operator `lazy`, which acts as `function`; moreover, it does wrap around the `function` operator as expected. See Figure (1) for an example.

Extra work is needed to handle `&optional` and `&key` parameters, but the overall design lies in this tweaking of the calling point and in allowing lazy functional objects to be passed around as regular functions (of course to be called via `lazy:call`).

2.1 Example: Lazy Functional Conses

Another example which turns out to be more easily realizable with CLAZY is the standard “*conses are functions*” one.

```
(deflazy conc (head tail)
  (lambda (selector)
    (ecase selector
      (car head)
      (cdr tail))))

(deflazy head (cons) (funcall cons 'car))

(deflazy tail (cons) (funcall cons 'cdr))
```

Now, we can build truly lazy lists²

```
CL prompt> (defparameter ll
             (lazy:call 'conc
                       1
                       (lazy:call 'conc
                                   (loop)
                                   (lazy:call 'conc
                                             3
                                             (loop)))))

LL
```

```
CL prompt> (head (tail (tail ll)))
3
```

Or the usual streams from SICP as the *integers* here below.

```
(defun integers-from (n)
  (lazy:call 'conc n (integers-from (1+ n))))

(defparameter integers (integers-from 0))
```

Yet, it must be noted that having normal order evaluation at one’s disposal naturally leads to the implementation of much more complex and sophisticated functional software, as in the case of the integrators in Section 3.5 of [1].

²Note where the (loop) calls appear.

```

(lazy:call (lazy (lambda (x &key (y (loop) y-supplied-p))
                (if y-supplied-p y (+ x 21))))
  21)

(lazy:call (lazy (lambda (x &key ((:y yy) (loop)))
                (if x (+ x 21) yy)))
  21)

(lazy:call (lazy (lambda (x &key ((:y yy) (loop)))
                (if x (+ x 21) yy)))
  nil :y 42))

```

Figure 2: `&key` arguments are dealt with as expected. The answer is always, as expected, 42.

2.2 Extra Considerations

CLAZY is supposed to be used in a very controlled way. While it is true that it adds *normal order evaluation* to Common Lisp, the user must remember that s/he is not using Haskell or a similar language. At its core, Common Lisp is a *strict* language, which allows side-effects; not a good mix to produce lazy code in a careless way. See also the note on *normal order evaluation* in Section 3.5 on streams of [1].

3 Reference Implementation

The CLAZY reference implementation can be found at common-lisp.net. The implementation lies within a package nicknamed LAZY and is based on the macros `lazy:call`, `lazy:deflazy`, and `lazy:lazy`.

The `lazy:call` macro is used at calling time (as the name implies). The `deflazy` macro is used to define functions. The `lazy` “special operator” returns a functional object that should be called in a lazy way, although the system is set up in such a way to “pass through” constant values (as tested by `constanp`).

The reference implementation is based on the pre-processing of lambda list arguments by `deflazy`: each argument is substituted by an internal name, which is expected to be bound to a *thunk* generated by `lazy:call` as per `delay`. In the body of a lazy function (or of a *lazy lambda*) each lambda list argument is actually re-defined as a `symbol-macrolet`, which expands in the appropriate `force` call. `deflazy` installs the lazy version of the function being defined in the property list of the function name.

Ordinary Lambda List Processing. As noted before, CLAZY pre-processes `&optional` and `&key` arguments in such a way to preserve the expected Common Lisp semantics. E.g., the calls in Figure (2) yield 42 as expected. On the contrary, the implementation does not treat `&rest` arguments in a special way (i.e., they are not *thunked*), this is because there is no way to access the list forming machinery in Common Lisp when `&rest` arguments are present; in a lazy piece of code, the list in the `&rest` argument will contain the actual thunks generated as if by `delay`.

4 Conclusions

CLAZY is an a exercise in Common Lisp style, which is also useful. The CLAZY library shows how, at the price of introducing a special call operator (`lazy:call`), it is possible to introduce *normal order* or *lazy* evaluation in Common Lisp. The extension has the following desirable characteristics: (i) it does not require the construction of a full blown interpreter implementing lazy evaluation, and (ii) thanks to the `deflazy` macro it allows a programmer to write code in the most natural way. It is much more difficult to achieve the same effect in any other language than Common Lisp, even when the language has macros. It is the under-the-hood interaction of macros and `symbol-macrolet` that makes CLAZY possible.

Of course, once this basic machinery is in place, extra Common Lisp incantations can be made and reader macros put in place as desired.

CLAZY is not perfect of course. The main open issue to complete the integration within the frame provided by Common Lisp is to work out a way to deal with CLOS methods. One way to achieve this would be to automatically define a method specializing on thunks for a given generic function. While this may work, it does open up typing issues³ that need to be worked out in details before proceeding with a full blown proposal.

References

- [1] H. Abelson, J. Sussman, and J. Sussman. *Structure and Interpretation of Computer Programs*. MIT Press, second edition, 1996.
- [2] A. Baine. Funds: Functional Data Structures in Common Lisp. Project page at <http://common-lisp.net/project/funds>, 2007.
- [3] S Burson. FSet: a functional set-theoretic collections library. Project page at <http://common-lisp.net/project/fset>, 2007.
- [4] M. Lamari. Heresy: Haskell-esque lazy-list and functional tools with a Common Lisp slant. Project page at <http://cl-heresy.sourceforge.net/Heresy.htm>, 2007.
- [5] X. Leroy. The OCaml system release 4.02. Published online at <http://caml.inria.fr/pub/docs/manual-ocaml/>, September 2014.
- [6] S. Marlow (editor). Haskell 2010 Language Report. Published online at: <https://www.haskell.org/onlinereport/haskell2010/>, November 2009.
- [7] R. Milner, M. Tofte, R. Harper, and D. MacQueen. *The Definition of Standard ML (Revised)*. MIT Press, 1997.
- [8] C. Okasaki. *Purely Functional Data Structures*. Cambridge University Press, 1998.
- [9] K. M. Pitman. The Common Lisp Hyperspec. Published online at <http://www.lisp.org/HyperSpec/FrontMatter/index.html>, 1994.

³`lazy:call` would need to know the actual resulting type of the argument expressions to meaningfully set up a discrimination for the underlying method.